

```

*****;
* RAMMETILSKUDD TIL KOMMUNENE 2005. RNB ;
* PROGRAMMET BEREGNER Revidert BUDSJETT 2005 ;
* HAR ENDRET UB OG KAP 571 I.F.T. IS05.IS05 ;
* PROGRAMMET BRUKES I UTBETALINGSPROGRAMMET FOR 2005 ;
* -----;
* Programmet beregner/henter tall til GH 2005 i denne rekkefølge ;
* ;
* 1) UTGIFTSUTJEVNING ;
* -henter kriteriedata og beregner indekser ;
* 2) TABELL B SAKER ;
* -overgangsordning 2005 ;
* -retter overgangsordning 2004 ;
* 3) TABELL C SAKER ;
* 4) INNBYGGERTILSKUDD ;
* 5) RAMMETILSKUDD I ALT ;
* 6) TABELLER ;
* ;
*****;

```

```

titlel 'Beregnet rammetilskudd for 2005';

```

```

options nonumber;
options date;
options pagesize=500;
options symbolgen;

```

```

%LET deflator=1.027;

```

```

* saldert budsjett 2005: ;
* - rammeoverføringer øker med 640 millioner kroner (UB effekt) ;
* - rammetilskudd øker med 492 mill. for å kompensere reduksjon av skatt (ikke UB effekt); ;
* ***** ;
* revidert budsjett 2005: ;
* - skjønnsmidlene øker med 70 millioner (pga. moms) (ikke UB effekt) ;
* - det trekkes ut 75 millioner kroner pga. feil ved gammel moms ikke UB effekt) ;
* - rammetilskudd øker med 147 mill. for å kompensere reduksjon av skatt (ikke UB effekt); ;

```

```

%let ub = (111672000 + 640000); * kommunenes utgiftsbehov;
%let kap571 = (34973075 + 640000 + 492000 + 70000 - 75506 + 147000); * sum rammetilskudd;
%let selskapskatt = 5486357; * selskapskatt, del av kap 571;
%let skjonn = (2772285 + 70000); * sum skjønnstilskudd;
%let ufordelt = 338821; * ufordelt skjønnstilskudd FM;
%let tilbakeholdt = 167000; * tilbakeholdt skjønn av KRd;
%let momsskjonn = 70000; * ekstraskjønn for komp. av moms;

```

```

* * * * *
* * * 1. UTGIFTSUTJEVNING * * * *
* * * * *;

```

```

*****;
* henter landbrukskriteriene fra salderingsprogrammet ;
* leser de øvrige kriteriedata inn her ;

```

```

*****;

data kriteriedata;
  merge is05.krit05 (rename= (k0_5=k0_5x k6_15=k6_15x k16_66=k16_66x k67_79=k67_79x
                             k80_89=k80_89x k89over=k89overx sumk0_5=sumk0_5x
                             sumk6_15=sumk6_15x sumk16_66=sumk16_66x sumk67_79=sumk67_79x
                             sumk80_89=sumk80_89x sumk89over=sumk89overx))
        is05.is05k_saldert (keep = knr landbr)
        ;
  by knr;

*****;
****      Lager totalsummer for kriteriedata.      ****;
****      Befolkningstall er summert i is05.krit05  ****;
*****;

PROC means DATA=kriteriedata noprint;
  VAR urbkrit kskilte kledig kdode ikkegift kinnv
      reisetid sone nabo pu0_15 pul6_99 landbr s_trekk vertspu;

  OUTPUT OUT= sumfill(DROP = _type_ _freq_)
          SUM= sumurbkrit sumkskilte sumkledig sumkdode sumikkegift
              sumkinnv sumreisetid sumsone sumnabo
              sumpu0_15 sumpul6_99 sumlandbr ss_trekk sumvertspu;

* leser inn befolkningstall per 1.1.05, er allerede summert i is05.knybef05 ;

DATA kritsum;
  if _N_=1 then set sumfill (obs=1);
  merge kriteriedata
        is05.knybef05
        ;
  by knr;

*****;
* beregner indekser for kriterier og for utgiftsbehovet ;
* med ikke oppdaterte kriteriedata (som i saldert)      ;
* dette bare til kontroll                               ;
*****;

data utgift_gml;
  set kritsum ;

  nn04 = k04 / sumk04;
  nn04GH = nn04 * 1000;
  basiskrit = 1;

  n105 = 1/433/nn04;          * basis;
  n204x = k0_5x/sumk0_5x/nn04; * 0-5;
  n304x = k6_15x/sumk6_15x/nn04; * 6-15;
  n404x = k16_66x/sumk16_66x/nn04; * 16-66;
  n504x = k67_79x/sumk67_79x/nn04; * 67-79;
  n604x = k80_89x/sumk80_89x/nn04; * 80-89;
  n704x = k89overx/sumk89overx/nn04; * over 90;
  n805 = kskilte/sumkskilte/nn04; * skilte;
  n905 = kledig/sumkledig/nn04; * ledig;
  n1005 = reisetid/sumreisetid/nn04; * reisetid;

```

```

n1705 = sone/sumsone/nn04;          * sone (nytt bosettingskriterium i 2002);
n1805 = nabo/sumnabo/nn04;         * nabo (nytt bosettingskriterium i 2002);
n1205 = kdode/sumkdode/nn04;       * dødelighet;
n1305 = ikkegift/sumikkegift/nn04; * ikke gifte;
n1405 = kinnv/sumkinnv/nn04;       * innvandrere;
n1505 = pu16_99/sumpu16_99/nn04;   * pu over 16;
n1605 = pu0_15/sumpu0_15/nn04;     * pu under 16;
n1905 = urbkrit/sumurbkrit/nn04;    * urbanitet;
n2005 = landbr/sumlandbr/nn04;      * landbruk;

```

```

label n105='Indeks basiskriteriet'
n204x='Indeks 0-5 åringer (04-tall)'
n304x='Indeks 6-15 åringer (04-tall)'
n404x='Indeks 16-66 åringer (04-tall)'
n504x='Indeks 67-79 åringer (04-tall)'
n604x='Indeks 80-89 åringer (04-tall)'
n704x='Indeks 89 åringer og over (04-tall)'
n805='Indeks skilte'
n905='Indeks arbeidsledige'
n1005='Indeks reisetid'
n1705='Indeks sone'
n1805='Indeks nabo'
n1205='Indeks dødlighet'
n1305='Indeks ikke gifte'
n1405='Indeks innvandrere'
n1505='Indeks PU over 16'
n1605='Indeks PU under 16'
n1905='Indeks urbanitet'
n2005='Indeks landbruk'
nn04='Andel befolkning (04-tall)'
nn04GH='Andel befolkning (04-tall), promille'
;

```

```

*-----;
*      Indeks for utgiftsbehovet      ;
*-----;

```

```

ny_gml =  n105    * 0.025
          + n204x  * 0.023
          + n304x  * 0.307
          + n404x  * 0.121
          + n504x  * 0.085
          + n604x  * 0.133
          + n704x  * 0.049
          + n805   * 0.038
          + n905   * 0.011
          + n1005  * 0.015
          + n1705  * 0.010
          + n1805  * 0.011
          + n1205  * 0.025
          + n1305  * 0.025
          + n1405  * 0.005
          + n1505  * 0.066
          + n1605  * 0.004
          + n1905  * 0.042
          + n2005  * 0.005;

```

```

nykostn_gml = ny_gml * nn04;
nykostn_gmlGH = nykostn_gml * 1000;

label ny_gml='Indeks ber. utgiftsbehov (04-tall)'
nykostn_gml='Andel ber. utgiftsbehov (04-tall)'
nykostn_gmlGH='Andel ber. utgiftsbehov (04-tall), promille'
;

* Beregner andeler for hvert kriterium til kontroll;

nyk105 = 1000* n105 * nn04;
nyk204x = 1000* n204x *nn04;
nyk304x = 1000* n304x *nn04;
nyk404x = 1000* n404x *nn04;
nyk504x = 1000* n504x *nn04;
nyk604x = 1000* n604x *nn04;
nyk704x = 1000* n704x *nn04;
nyk805 = 1000* n805 *nn04;
nyk905 = 1000* n905 *nn04;
nyk1005 = 1000* n1005 *nn04;
nyk1705 = 1000* n1705 *nn04;
nyk1805 = 1000* n1805 *nn04;
nyk1205 = 1000* n1205 *nn04;
nyk1305 = 1000* n1305 *nn04;
nyk1405 = 1000* n1405 *nn04;
nyk1505 = 1000* n1505 *nn04;
nyk1605 = 1000* n1605 *nn04;
nyk1905 = 1000* n1905 *nn04;
nyk2005 = 1000* n2005 *nn04;

label nyk105='Andel basiskriteriet'
nyk204x='Andel 0-5 åringer (04-tall)'
nyk304x='Andel 6-15 åringer (04-tall)'
nyk404x='Andel 16-66 åringer (04-tall)'
nyk504x='Andel 67-79 åringer (04-tall)'
nyk604x='Andel 80-89 åringer (04-tall)'
nyk704x='Andel 89 åringer og over (04-tall)'
nyk805='Andel skilte'
nyk905='Andel arbeidsledige'
nyk1005='Andel reisetid'
nyk1705='Andel sone'
nyk1805='Andel nabo'
nyk1205='Andel dødlighet'
nyk1305='Andel ikke gifte'
nyk1405='Andel innvandrere'
nyk1505='Andel PU over 16'
nyk1605='Andel PU under 16'
nyk1905='Andel urbanitet'
nyk2005='Andel landbruk'
;

*****;
* beregner indekser for utgiftsutjevning og andeler med nye befolkningsgtall;
* indekser som ikke oppdateres hentes inn fra beregningen ovenfor ;
*****;

data utgift_rnb;

```

```

merge utgift_gml (keep= knr nn04 nn04GH n105 n805 n905 n1005 n1705 n1805 n1205
                    n1305 n1405 n1505 n1605 n1905 n2005 k05 k0_5
                    k6_15 k16_66 k67_79 k80_89 k89over sumk05 sumk0_5
                    sumk6_15 sumk16_66 sumk67_79 sumk80_89 sumk89over
                    k04 sumk04 basiskrit kskilte kledig reisetid sone nabo
                    kdode ikkegift kinnv pul6_99 pu0_15 urbkrit landbr)
                    is05.is05k_saldert (keep= knr s_trekkGH st_pra st_pr05 areal denheter eiendommer
                                         dmark)
;
by knr;

nn05 = k05 / sumk05;
nn05GH = nn05 * 1000;

* beregner andeler for alderskriterier med ny bef.tall ;

nyk205 = k0_5/sumk0_5;
nyk305 = k6_15/sumk6_15;
nyk405 = k16_66/sumk16_66;
nyk505 = k67_79/sumk67_79;
nyk605 = k80_89/sumk80_89;
nyk705 = k89over/sumk89over;

label nyk205='Andel 0-5 åringer (05-tall)'
nyk305='Andel 6-15 åringer (05_tall)'
nyk405='Andel 16-66 åringer (05-tall)'
nyk505='Andel 67-79 åringer (05-tall)'
nyk605='Andel 80-89 åringer (05-tall)'
nyk705='Andel 89 åringer og over (05-tall)'
;

* beregner indekser for alderskriterier med ny bef.tall ;

n205 = k0_5/sumk0_5/nn05;          * 0-5;
n305 = k6_15/sumk6_15/nn05;       * 6-15;
n405 = k16_66/sumk16_66/nn05;     * 16-66;
n505 = k67_79/sumk67_79/nn05;     * 67-79;
n605 = k80_89/sumk80_89/nn05;     * 80-89;
n705 = k89over/sumk89over/nn05;    * over 90;

label n205='Indeks 0-5 åringer (05-tall)'
n305='Indeks 6_15 åringer (05-tall)'
n405='Indeks 16-66 åringer (05-tall)'
n505='Indeks 67-79 åringer (05-tall)'
n605='Indeks 80-89 åringer (05-tall)'
n705='Indeks 89 åringer og over (05-tall)'
nn05='Andel befolkning (05-tall)'
nn05GH='Andel befolkning (05-tall), promille'
;

*****;
* beregner de to delindeksene, vekt er summen av kriterievektene ;
* som inngår i de to forskjellige delindeksene ;
*****;

%let vekt = (0.023 + 0.307 + 0.121 + 0.085 + 0.133 + 0.049);
* vekt er nå 0,718 ;

```

```

ny_bef = (n205 * 0.023 +
          n305 * 0.307 +
          n405 * 0.121 +
          n505 * 0.085 +
          n605 * 0.133 +
          n705 * 0.049) / &vekt.;

ny_rest = (n105 * 0.025 +
           n805 * 0.038 +
           n905 * 0.011 +
           n1005 * 0.015 +
           n1705 * 0.010 +
           n1805 * 0.011 +
           n1205 * 0.025 +
           n1305 * 0.025 +
           n1405 * 0.005 +
           n1505 * 0.066 +
           n1605 * 0.004 +
           n1905 * 0.042 +
           n2005 * 0.005) / (1 - &vekt.);

nykostn_bef = &vekt. * ny_bef * nn05;
nykostn_rest = (1 - &vekt.) * ny_rest * nn04;
nykostn_rnb = nykostn_bef + nykostn_rest;
nykostn_rnbGH = nykostn_rnb * 1000;

label ny_bef='Delindeks ber. utgiftsbehov, befolkning'
      ny_rest='Delindeks ber. utgiftsbehov, resterende kriterier'
      nykostn_bef='Andel ber. utgiftsbehov, befolkning'
      nykostn_rest='Andel ber. utgiftsbehov, resterende kriterier'
      nykostn_rnb='Andel ber. utgiftsbehov, vektet sammen'
      nykostn_rnbGH='Andel ber. utgiftsbehov, vektet sammen (i promille)'
;

*****;
****      BEREGNER UTGIFTSUTJEVNING      ****;
*****;

* for befolkningskriteriene ;

UBPI_bef = &vekt. * &ub. * 1000 / sumk05;
UBPIK_bef = UBPI_BEF * ny_bef;
OMFPI_bef = UBPIK_bef - UBPI_bef;
OMF_bef = (OMFPI_bef * k05) / 1000;

* for de øvrige kriteriene ;

UBPI_rest = (1 - &vekt.) * &ub. * 1000 / sumk04;
UBPIK_rest = UBPI_rest * ny_rest;
OMFPI_rest = UBPIK_rest - UBPI_rest;
OMF_rest = (OMFPI_rest * k04) / 1000;

* total omfordeling ;

OMFX_rnb = OMF_bef + OMF_rest;

```

```

OMF05_rnb = OMF_X_rnb + st_pr05;

label UBPI_bef='Gj. ber. utgiftsbehov, befolkningskriteriene'
      UBPIK_bef='Ber. utgiftsbehov, befolkningskriteriene (05-tall)'
      OMFPI_bef='Omfordeling per innbygger, befolkningskriteriene (05-tall)'
      OMF_bef='Omfordeling, befolkningskriteriene (05-tall), 1000 kr.'
      UBPI_rest='Gj. ber. utgiftsbehov, resterende kriterier'
      UBPIK_rest='Ber. utgiftsbehov, resterende kriterier'
      OMFPI_rest='Omfordeling per innbygger, resterende kriterier'
      OMF_rest='Omfordeling, resterende kriterier (1000 kr.)'
      OMF_X_rnb='Samlet omfordeling utgiftsutjevning (1000 kr.)'
      St_pr05='Netto effekt korreksjonsordning statlig/privat skole (1000 kr.)'
      OMF05_rnb='Total omfordeling utg.utj. og korreksjon (1000 kr.)'
;

* * * * * ;
* * *      2.      TABELL B SAKER      * * * * * ;
* * * * * ;

*****;
* - Beregner overgangsordning 2005      ;
* - Beregner overgangsordning i alt      ;
*****;

*****;
* I RNB 05 legges den nye kostnadsnøkkele til      ;
* grunn for overgangsordningen i tabell B      ;
* det er ikke noen nye saker i tabell B      ;
*****;

data tabB_grunnlag;
  merge is05.is05k_saldert (keep= knr elavgift vertskom mangfold
                           timetall_B skatteoppkrever krise vertsknokkal
                           korreksj ovg04_6 ovg04_moms_6 ovg03_4 ovg02_2)
        utgift_rnb (keep= knr nykostn_rnb)
        ;
  by knr;

*****;
* seks endringer i tab. B fordeles etter kostnadsnøkkelen      ;
* beregne ny fordeling etter oppdatert kostnadsnøkkel      ;
*****;

vertsknokkal_rnb = (520551-84370) * nykostn_rnb;
krise_rnb = -28779 * nykostn_rnb;
timetall_B_rnb = 255000 * nykostn_rnb;
flere_elever_rnb = -50200 * nykostn_rnb;
skatteoppkrever_rnb = -14500 * nykostn_rnb;
mangfold_rnb = 3709 * nykostn_rnb;

*****;
* Korreksjonene i tabell B fordeler seg etter      ;
* den nye kostnadsnøkkeln      ;
*****;

sum6_rnb = vertsknokkal_rnb + krise_rnb + timetall_b_rnb + flere_elever_rnb +

```

```

skatteoppkrever_rnb + mangfold_rnb;

korreksj_rnb = vertskom + elavgift + vertsknokkel_rnb + krise_rnb + timetall_b_rnb +
flere_elever_rnb + skatteoppkrever_rnb + mangfold_rnb;

label flere_elever_rnb='Uttrekk pga. flere elever i privat/statlig skole, ny k-nøkkel'
sum6_rnb='Summen av 6 endringer som fordelse etter k-nøkkel'
korreksj_rnb='Summen av alle korreksjoner i 2005 (inngående)'
;

proc means data=tabB_grunnlag noprint nway;
var korreksj_rnb;

output out= sumtabB
sum= sumkorr
;

data tabB_rnb;
if _n_ = 1 then set sumtabB;
set tabB_grunnlag;

korrkost_rnb = sumkorr * nykostn_rnb;

*****;
* Differanse inn.fordeling- fordeling etter kostnadsnøkkelen ;
*****;

ovg05_rnb = korreksj - korrkost_rnb;
ovg05_8_rnb = ovg05_rnb * 0.8;

*****;
* OBS. Rettes i rundskriv til høsten. Riktig variabel er ;
* korreksj_rnb ;
*****;

*****;
* beregner en ny samlet overgangsordning for 2005 ;
*****;

sovg05_rnb = ovg05_8_rnb + ovg04_6 + ovg03_4 + ovg02_2;

label korrkost_rnb='Korreksjoner 2005, fordelt med k-nøkkel'
ovg05_rnb='Overg. korreksjoner RNB 2005'
ovg05_8_rnb='1. års virkning av overg. korreksjoner RNB 2005'
sovg05_rnb='Samlet overgangsordning RNB 2005 (1000 kr.)'
;

* * * * * ;
* * * * 3. TABELL C SAKER * * * * ;
* * * * * ;

*****;
* 1) TIMETALL_C = Økt timetall grunnskolen 2004/2005 ;
* - Fordeles HER etter andel 6-15 åringer ;

```



```

* - Beløp = 130262 (2004-kr) som skal prisjusteres ;
* 2) MOMSKOMP C = momskompensasjon engangsuttrekk ;
* - Særskilt fordeling i is05.tabB_C ;
* - Beløp -1580097 (2004 kr) som skal prisjusteres ;
* - Er allerede prisjustert i is05.tabB_C ;
* 3) OPPGAVE = Forsøk med oppgavedifferensiering ;
* - Særskilt fordeling i is05.tabB_C ;
* - Prisjustering av tall fra RNB2004 ;
* 4) CSS = Conrad Svendsen Senter ;
* - Særskilt fordeling i is05.tabB_C ;
* - Prisjustering av tall fra RNB2004 ;
* 5) INNDEL05 = Inndelingstilskudd ;
* - Tallene ferdigstilt i is05.tabB_C ;
* - Prisjustert 2004-tilskuddet til RE (Våle+Ramnes) ;
* Tatt inn nytt tilskudd til Bodø (Bodø+Skjerstad) ;
* 6) KORRMOMS = korreksjon uttrekk moms (ny i RNB 05) ;
* - 75 millioner kroner trekkes ut fra kommunene (engang) ;
* - særskilt fordeling leses inn her ;
*****;

```

```

data korrmoms;
  infile 'g:\is\is05\revidert grønt hefte 2005\korreksjoner og innlemminger\
moms korreksjon kommuner.csv' missover dlm=";" firstobs=7 obs=439;
  input knr knavn $ korrmoms;

```

```

  label korrmoms='Korreksjon av uttrekk momskompensasjon 2005';

```

```

data tiltabC;
  merge is05.is05k_saldert (keep=knr oppgave CSS inndel05 momskompC
                           timetall_C vertspu_i_tabC)
        korrmoms
        ;
  by knr;

  utenpaa_rnb = CSS + oppgave + inndel05 + timetall_C + momskompC + korrmoms;

```

```

PROC means DATA=tiltabC noprint;
  VAR utenpaa_rnb;

  OUTPUT OUT= sumfill (DROP = _type_ _freq_)
          SUM= sumutenpaa_rnb
          ;

```

```

DATA tabC_rnb;
  if _N_=1 then set sumfill (obs=1);
  set tiltabC;

  label utenpaa_rnb='Sum tabell C, RNB 05 (1000 kr.)';

```

```

* * * * * ;
* * *      4. INNBYGGERTILSKUDD      * * * ;
* * * * * ;

*****;
* Samler Tabell C, Tabell B, Utgiftsutjevning, Kriteriedata ;
* Skjønns- Regional og Nord-norge tilskudd+rammetilsk 2003-4 ;

```

```

* Og tar inn selskapskatt (fra skatteåret 2002) ;
*****;

data rnb05k;
merge is05.is05k_saldert (keep = knr knavn skattpst03 oskjon05 sumoskjon05
                           xskjon05 sumxskjon05 reg05 sumreg05
                           nntil05 sumnntil05 selskap_02)
      utgift_rnb
      tabB_rnb
      tabC_rnb;
by knr;

fylke = Int(knr / 100);

*****      SKJØNNSTILSKUDD      *****;
*   SKJØNN i alt          = 2 772 285 + 70 000 ;
*   Ordinært              = 1 968 785          ;
*   Ekstraordinært       =   297 679          ;
*   Ufordelt FM           =   338 821          ;
*   Tilbakeholdt KRD     =   167 000          ;
*   Ekstraskjønn moms    =    70 000          ;
*****;

Ramme60_rnb = &kap571. - &selskapskatt. - sumreg05 - sumnntil05 -
              sumxskjon05 - sumoskjon05 - &ufordelt. - &tilbakeholdt. -
              &momsskjon.;

*****;
* INNBYGGERTILSKUDD FØR UTJEVNING 2005 ;
*****;

pcapx_rnb = (ramme60_rnb - sumutenpaa_rnb) * 1000 / sumk05;
pcap05_rnb = (pcapx_rnb / 1000 * k05);

*****;
*   POST 60 (p60_05_rnb) ;
*-----;
* Inklusiv Utgififtsutjevning, overgangsordning og tab C saker;
*****;

p60_05_rnb = round(pcap05_rnb + OMF05_rnb + sovg05_rnb + utenpaa_rnb);
if knr=301 then p60_05_rnb = p60_05_rnb + 3; * Avrundingsfeil belastes Oslo;

* * * * *
* * *   5.  RAMMETILSKUDD I ALT   * * * * *
* * * * *

Fastdel_rnb = p60_05_rnb + nntil05 + reg05;
Sumskjon = xskjon05 + oskjon05;
Rammetil_rnb05 = Fastdel_rnb + Sumskjon;
rammetil_rnb05_sel = Rammetil_rnb05 + selskap_02;

label ramme60_rnb='Bevilgninger på post 60, RNB 2005 (1000 kr.)'
      pcapx_rnb='Innbyggertilskudd før utjevning, RNB 05 (kr. per innb.)'
      pcap05_rnb='Innbyggertilskudd før utjevning, RNB 05 (1000 kr.)'
      p60_05_rnb='Innbyggertilskudd etter utjevning, RNB 05 (1000 kr.)'

```

```

fastdel_rnb='Fast del, RNB 05 (1000 kr.)'
sumskjon='Sum skjønnstilskudd (fordelt kommunevis), RNB 05 (1000 kr.)'
rammetil_rnb05='Rammetilskudd eksl. selskapsskatt, RNB 05 (1000 kr.)'
rammetil_rnb05_sel='Rammetilskudd inkl. selskapsskatt, RNB 05 (1000 kr.)'
;

```

```

*****;
* OBS ;
* Tilbakeholdt og ufordelt skjønn settes inn manuelt i tabellene ;
* dette gjelder også skjønnsøkningen av 70 millioner til moms ;
*****;

```

```

*****;
* Lagret permanent på is05.rnb05k av MSC 21.06.2005 ;
*****;

```

```

Data rnb05k;
set rnb05k;

```

```

* * * * * ;
* * * * 6. TABELLER * * * * ;
* * * * * ;

```

```

*****;
* lagre fylkessummene som en egen datafil for (bl.a.) tabell E ;
*****;

```

```

proc means data=is05.rnb05k noprint nway;
class fylke;
var k04 k05;

```

```

output out=fylkesum
sum=fysumk04 fysumk05
;

```

```

data fysum;
merge is05.rnb05k
fylkesum
;

```

```
by fylke;
```

```
if fylke = . then delete;
```

```

fyl_befandel04 = k04 / fysumk04;
fyl_befandel05 = k05 / fysumk05;

```

```

fyl_ny_bef = ny_bef * fyl_befandel05;
fyl_ny_rest = ny_rest * fyl_befandel04;

```

```

fyl_ubpi_bef = ubpi_bef * fyl_befandel05;
fyl_ubpi_rest = ubpi_rest * fyl_befandel04;

```

```

fyl_ubpik_bef = ubpik_bef * fyl_befandel05;
fyl_ubpik_rest = ubpik_rest * fyl_befandel04;

```

```

fyl_omfpi_bef = omfpi_bef * fyl_befandel05;
fyl_omfpi_rest = omfpi_rest * fyl_befandel04;

fyl_n105 = n105 * fyl_befandel04;
fyl_n205 = n205 * fyl_befandel05;
fyl_n305 = n305 * fyl_befandel05;
fyl_n405 = n405 * fyl_befandel05;
fyl_n505 = n505 * fyl_befandel05;
fyl_n605 = n605 * fyl_befandel05;
fyl_n705 = n705 * fyl_befandel05;
fyl_n805 = n805 * fyl_befandel04;
fyl_n905 = n905 * fyl_befandel04;
fyl_n1005 = n1005 * fyl_befandel04;
fyl_n1705 = n1705 * fyl_befandel04;
fyl_n1805 = n1805 * fyl_befandel04;
fyl_n1205 = n1205 * fyl_befandel04;
fyl_n1305 = n1305 * fyl_befandel04;
fyl_n1405 = n1405 * fyl_befandel04;
fyl_n1505 = n1505 * fyl_befandel04;
fyl_n1605 = n1605 * fyl_befandel04;
fyl_n1905 = n1905 * fyl_befandel04;
fyl_n2005 = n2005 * fyl_befandel04;

label fyl_ny_bef='Indeks beregnet utg.behov, befolkning'
      fyl_ny_rest='Indeks beregnet utg.behov, rest'
      fyl_ubpi_bef='Gj. utgiftsbehov, befolkning'
      fyl_ubpi_rest='Gj. utgiftsbehov, rest'
      fyl_ubpik_bef='Beregnet utgiftsbehov, befolkning'
      fyl_ubpik_rest='Beregnet utgiftsbehov, rest'
      fyl_omfpi_bef='Omfordeling per innb., befolkning'
      fyl_omfpi_rest='Omfordeling per innb., rest'
;

*****;
*UTSKRIFTER ;
*****;

* tabell 1;

ods html file= 'g:\is\is05\revidert grønt hefte 2005\tabeller kom\tabell 1.html';
proc tabulate data=is05.rnb05k f=20.8;
class knr fylke;
var p60_05_rnb nntil05 reg05 fastdel_rnb xskjon05 oskjon05 sumskjon
    rammetil_rnb05 selskap_02 rammetil_rnb05_sel;
format fylke fylkenn. knr kom05no.;
table fylke=' ' * (knr=' ' all=' ') all='hele landet',
      (p60_05_rnb nntil05 reg05 fastdel_rnb xskjon05 oskjon05 sumskjon
      rammetil_rnb05 selskap_02 rammetil_rnb05_sel) * sum=' '
      /rts=35;
title 'Tabell 1: Rammetilskudd til kommunene 2005.';
run;
ods html close;
run;

* Tabell 2;

```

```
ods html file= 'g:\is\is05\revidert grønt hefte 2005\tabeller kom\tabell 2.html';
proc tabulate data=is05.rnb05k f=20.8;
class knr fylke;
var pcap05_rnb omf05_rnb sovg05_rnb utenpaa_rnb p60_05_rnb;
format fylke fylkenn. knr kom05no.;
table fylke=' ' * (knr=' ' all='') all='hele landet',
      (pcap05_rnb omf05_rnb sovg05_rnb utenpaa_rnb p60_05_rnb) * sum=' '
      /rts=35;
title 'Tabell 2: Innbyggertilskudd for kommunene 2005.';
run;
ods html close;
run;
```

* Tabell A;

```
ods html file= 'g:\is\is05\revidert grønt hefte 2005\tabeller kom\tabell A.html';
proc tabulate data=is05.rnb05k f=20.8;
class knr fylke;
var ny_bef ubpi_bef ubpik_bef omfpi_bef omf_bef ny_rest ubpi_rest
      ubpik_rest omfpi_rest omf_rest omfx_rnb s_trekkGH st_pra st_pr05 omf05_rnb;
format fylke fylkenn. knr kom05no.;
table fylke=' ' * (knr=' ' all='') all='hele landet',
      (ny_bef ubpi_bef ubpik_bef omfpi_bef omf_bef ny_rest ubpi_rest ubpik_rest
      omfpi_rest omf_rest omfx_rnb s_trekkGH st_pra st_pr05 omf05_rnb) * sum=' '
      /rts=35;
title 'Tabell A: Utgiftsutjevningen mm. Kommunene 2005.';
run;
ods html close;
run;
```

```
proc means data=fysum noprint nway;
class fylke;
var fyl_ny_bef fyl_ubpi_bef fyl_ubpik_bef fyl_omfpi_bef omf_bef fyl_ny_rest fyl_ubpi_rest
      fyl_ubpik_rest fyl_omfpi_rest omf_rest omfx_rnb s_trekkGH st_pra st_pr05 omf05_rnb;

output out= fysum_taba
      sum= sumfyl_ny_bef sumfyl_ubpi_bef sumfyl_ubpik_bef sumfyl_omfpi_bef sumomf_bef
      sumfyl_ny_rest sumfyl_ubpi_rest sumfyl_ubpik_rest sumfyl_omfpi_rest
      sumomf_rest sumomfx_rnb sums_trekkGH sumst_pra sumst_pr05 sumomf05_rnb
      ;
```

```
ods html file='g:\is\is05\revidert grønt hefte 2005\tabeller kom\tab A fylkessummer.html';
proc print data=fysum_taba noobs label;
var fylke sumfyl_ny_bef sumfyl_ubpi_bef sumfyl_ubpik_bef sumfyl_omfpi_bef sumomf_bef
      sumfyl_ny_rest sumfyl_ubpi_rest sumfyl_ubpik_rest sumfyl_omfpi_rest sumomf_rest
      sumomfx_rnb sums_trekkGH sumst_pra sumst_pr05 sumomf05_rnb;
format fylke fylkenn.
      sumfyl_ny_bef sumfyl_ubpi_bef sumfyl_ubpik_bef sumfyl_omfpi_bef sumomf_bef
      sumfyl_ny_rest sumfyl_ubpi_rest sumfyl_ubpik_rest sumfyl_omfpi_rest sumomf_rest
      sumomfx_rnb sums_trekkGH sumst_pra sumst_pr05 sumomf05_rnb f20.8;
title 'Tabell A: Utgiftsutjevning. Fylkessummer for kommunene 2004';
run;
ods html close;
run;
```

```

data tabB;
  set is05.rnb05k;

  ovg04_rest_6 = ovg04_6 - ovg04_moms_6;

* Tabell B;

ods html file= 'g:\is\is05\revidert grønt hefte 2005\tabeller kom\tabell B.html';
proc tabulate data=tabB f=20.8;
class knr fylke;
var elavgift vertskom sum6_rnb korreksj_rnb nykostn_rnbGH korrkost_rnb
  ovg05_rnb ovg05_8_rnb ovg04_rest_6 ovg04_moms_6 ovg03_4 ovg02_2 sovg05_rnb;
format fylke fylkenn. knr kom05no.;
table fylke=' ' * (knr=' ' all=' ') all='hele landet',
  (elavgift vertskom sum6_rnb korreksj_rnb nykostn_rnbGH korrkost_rnb ovg05_rnb
  ovg05_8_rnb ovg04_rest_6 ovg04_moms_6 ovg03_4 ovg02_2 sovg05_rnb) * sum=' '
  /rts=35;
title 'Tabell B: Samlet overgangsordning. Kommunene 2005.';
run;
ods html close;
run;

* Tabell C;

ods html file= 'g:\is\is05\revidert grønt hefte 2005\tabeller kom\tabell C.html';
proc tabulate data=is05.rnb05k f=20.8;
class knr fylke;
var CSS oppgave inndel05 timetall_C momskompC korrmoms utenpaa_rnb;
format fylke fylkenn. knr kom05no.;
table fylke=' ' * (knr=' ' all=' ') all='Hele landet',
  (CSS oppgave inndel05 timetall_C momskompC korrmoms utenpaa_rnb) * sum=' '
  /rts=35;
title 'Tabell C: Utenfor overgangsordningen og inndelingstilskuddet. Kommunene 2004.';
run;
ods html close;
run;

* Tabell D;

ods html file= 'g:\is\is05\revidert grønt hefte 2005\tabeller kom\tabell D.html';
proc tabulate data=is05.rnb05k f=20.8;
class knr fylke;
var nn05GH n205 n305 n405 n505 n605 n705 ny_bef nn04GH n105 n1905 n805 n905
  n1005 n1705 n1805 n1205 n1305 n1405 n1505 n1605 n2005 ny_rest;
format fylke fylkenn. knr kom05no.;
table fylke=' ' * (knr=' ' all=' ') all='Hele landet',
  (nn05GH n205 n305 n405 n505 n605 n705 ny_bef nn04GH n105 n1905 n805 n905
  n1005 n1705 n1805 n1205 n1305 n1405 n1505 n1605 n2005 ny_rest) * sum=' '
  /rts=35;
title 'Tabell D: Indeks for beregnet utgiftsbehov. Kommunene 2005.';
run;
ods html close;
run;

```

```

proc means data=fysum noprint nway;
  class fylke;
  var nn05GH fyl_n205 fyl_n305 fyl_n405 fyl_n505 fyl_n605 fyl_n705
      fyl_ny_bef nn04GH fyl_n105 fyl_n1905 fyl_n805 fyl_n905 fyl_n1005 fyl_n1705
      fyl_n1805 fyl_n1205 fyl_n1305 fyl_n1405 fyl_n1505 fyl_n1605 fyl_n2005 fyl_ny_rest;

  output out=fysum_tabd
      sum=sum_nn05GH sum_n205 sum_n305 sum_n405 sum_n505 sum_n605
          sum_n705 sum_ny_bef sum_nn04GH sum_n105 sum_n1905 sum_n805 sum_n905 sum_n1005
          sum_n1705 sum_n1805 sum_n1205 sum_n1305 sum_n1405 sum_n1505 sum_n1605
          sum_n2005 sum_ny_rest;

ods html file='g:\is\is05\revidert grønt hefte 2005\tabeller kom\tab D fylkessummer.html';
proc print data=fysum_tabd noobs label;
  var fylke sum_nn05GH sum_n205 sum_n305 sum_n405 sum_n505 sum_n605
      sum_n705 sum_ny_bef sum_nn04GH sum_n105 sum_n1905 sum_n805 sum_n905 sum_n1005
      sum_n1705 sum_n1805 sum_n1205 sum_n1305 sum_n1405 sum_n1505 sum_n1605
      sum_n2005 sum_ny_rest;
  format fylke fylkenn.
      sum_nn05GH sum_n205 sum_n305 sum_n405 sum_n505 sum_n605 sum_n705 sum_ny_bef
      sum_nn04GH sum_n105 sum_n1905 sum_n805 sum_n905 sum_n1005 sum_n1705 sum_n1805
      sum_n1205 sum_n1305 sum_n1405 sum_n1505 sum_n1605 sum_n2005 sum_ny_rest f20.8
      ;
  title 'Tabell D: Indeks for beregnet utgiftsbehov. Fylkessummer for kommunene 2005';
run;
ods html close;
run;

*Tabell E;

data tabE;
  set is05.rnb05k;

  tabE_sone = sone / 1000;
  tabE_nabo = nabo / 1000;

ods html file= 'g:\is\is05\revidert grønt hefte 2005\tabeller kom\tabell E.html';
proc tabulate data=tabE f=20.8;
class knr fylke;
var k04 k05 basiskrit k0_5 k6_15 k16_66 k67_79 k80_89 k89over urbkrit
    kskilte kledig reisetid tabE_sone tabE_nabo kdode ikkegift kinnv
    pul6_99 pu0_15 areal denheter eiendommer dmark skattpst03;
format fylke fylkenn. knr kom05no.;
table fylke=' * (knr=' ' all='') all='hele landet',
    (k04 k05 basiskrit k0_5 k6_15 k16_66 k67_79 k80_89 k89over urbkrit
    kskilte kledig reisetid tabE_sone tabE_nabo kdode ikkegift kinnv
    pul6_99 pu0_15 areal denheter eiendommer dmark skattpst03)
    /rts=35;
title 'Tabell E : Kriteriedata 2005 - kommuner';
run;
ods html close;
run;

```